

Работа со сводными таблицами

В ЭТОЙ ГЛАВЕ...

- Ознакомительный пример
- Создание сложных сводных таблиц
- Создание нескольких сводных таблиц
- Создание обратной сводной таблицы

Эта глава посвящена сводным таблицам Excel — одной из важнейших тем, которые просто невозможно обойти стороной.

Ознакомительный пример

Сводные таблицы являются одним из самых мощных инновационных средств Excel. Они появились в Excel 5 и поддерживаются только в Excel (пока ни один программный продукт для управления электронными таблицами не предоставляет подобной возможности). В этой главе отсутствует информация об основах использования сводных таблиц. Предполагается, что вы уже знакомы с данной темой (как и со способами создания и изменения сводных таблиц вручную).

Создав сводную таблицу на основе базы данных или списка, можно быстро отобразить необходимые итоговые данные, которые другим способом представить на экране или бумаге просто невозможно. Кроме того, такое представление данных можно получить очень быстро и безо всяких формул. В Excel также поддерживается возможность создания кода VBA, который позволяет управлять сводными таблицами.

Этот раздел начинается с простого примера создания сводной таблицы средствами VBA.

На рис. 15.1 показана простая база данных рабочего листа (диапазон ячеек рабочего листа). Она включает следующие четыре поля: Торговый представитель, Регион, Месяц и Продажи. Каждая запись содержит сведения об объеме продаж торгового представителя за определенный месяц.



На сайте

Рассматриваемый далее пример находится на сайте книги (файл `simple pivot table.xlsx`).

	A	B	C	D
1	Торговый представитель	Регион	Месяц	Продажи
2	Эми	Север	Янв	33 488
3	Эми	Север	Фев	47 008
4	Эми	Север	Мар	32 128
5	Боб	Север	Янв	34 736
6	Боб	Север	Фев	92 872
7	Боб	Север	Мар	76 128
8	Чак	Юг	Янв	41 536
9	Чак	Юг	Фев	23 192
10	Чак	Юг	Мар	21 736
11	Дуг	Юг	Янв	44 834
12	Дуг	Юг	Фев	32 002
13	Дуг	Юг	Мар	23 932

Рис. 15.1. Эта таблица является хорошим кандидатом в сводные таблицы

Создание сводной таблицы

На рис. 15.2 показана сводная таблица, которая создана на основе указанных данных. Она обеспечивает отображение суммарного объема ежемесячных продаж каждого торгового представителя и содержит следующие поля.

- **Регион.** Поле фильтра отчета в сводной таблице.
- **Торговый представитель.** Поле строки в сводной таблице.
- **Месяц.** Поле столбца в сводной таблице.
- **Продажи.** Поле данных в сводной таблице, которое использует функцию Сумма (Sum).

	A	B	C	D	E	F	G	H	I	J	K	L
1	Регион	(Все)										
2												
3	Сумма по полю Продажи	Названия столбцов										
4	Названия строк	Янв	Фев	Мар	Общий итог							
5	Боб	34736	92872	76128	203736							
6	Дуг	44834	32002	23932	100768							
7	Чак	41536	23192	21736	86464							
8	Эми	33488	47008	32128	112624							
9	Общий итог	154594	195074	153924	503592							
10												
11												
12												
13												
14												
15												
16												
17												
18												
19												

Поля сводной таблицы

Выберите поля для добавления в отчет:

Торговый представитель

Регион

Месяц

Продажи

Перетащите поля в нужную область:

ФИЛЬТРЫ
 КОЛОННЫ

СТРОКИ
 Σ ЗНАЧЕНИЯ

Отложить обновление макета
 ОБНОВИТЬ

Рис. 15.2. Сводная таблица, созданная на основе таблицы, показанной на рис. 15.1

Перед созданием сводной таблицы была включена функция записи макроса. Созданная сводная таблица размещается на новом рабочем листе. Далее представлен автоматически сгенерированный код макроса.

```
Sub CreatePivotTable()  
  Sheets.Add  
  ActiveWorkbook.PivotCaches.Create _  
    (SourceType:=xlDatabase, _  
    SourceData:="Лист1!R1C1:R13C4", _  
    Version:=xlPivotTableVersion15).CreatePivotTable _  
    TableDestination:="Лист2!R3C1", _  
    TableName:="PivotTable1", _  
    DefaultVersion:=xlPivotTableVersion15)  
  Sheets("Лист2").Select  
  Cells(3, 1).Select  
  With ActiveSheet.PivotTables("PivotTable1") _  
    .PivotFields("Регион")  
    .Orientation = xlPageField  
    .Position = 1  
  End With  
  With ActiveSheet.PivotTables("PivotTable1") _  
    .PivotFields("Торговый представитель")  
    .Orientation = xlRowField  
    .Position = 1  
  End With  
  With ActiveSheet.PivotTables("PivotTable1") _  
    .PivotFields("Месяц")  
    .Orientation = xlColumnField  
    .Position = 1  
  End With  
  ActiveSheet.PivotTables("PivotTable1") _  
    .AddDataField ActiveSheet.PivotTables("PivotTable1") _  
    .PivotFields("Продажи"), "Сумма по полю Продажи", xlSum  
End Sub
```

Запуск записанного макроса скорее всего приведет к ошибке. В процессе проверки кода вы обнаружите, что функция записи макроса “жестко закодировала” название рабочего листа (Лист2), на котором создается сводная таблица. Если подобный лист уже существует (либо добавляется новый лист с другим именем), выполнение макроса приведет к ошибке.

Несмотря на неработоспособность записанного макроса, вряд ли его можно назвать совсем бесполезным. Разработчики кода VBA для сводных таблиц найдут в этом коде немало полезного для себя.

Просмотр кода VBA

Код VBA, записанный при создании сводной таблицы, может привести вас в замешательство. Для того чтобы разобраться в записанном макросе, потребуются кое-что знать об используемых объектах (соответствующая информация содержится в интерактивном справочном руководстве).

- `PivotCaches` — коллекция объектов `PivotCache` в объекте `Workbook` (данные для сводной таблицы, которые хранятся в кеше сводной таблицы).

- `PivotTables` — коллекция объектов `PivotTable` в объекте `Worksheet`.
- `PivotFields` — коллекция полей в объекте `PivotTable`.
- `PivotItems` — коллекция отдельных элементов данных в поле.
- `CreatePivotTable` — метод объекта `PivotCache`, который создает сводную таблицу на основе данных, содержащихся в кеш-памяти.



ВЫБОР ДАННЫХ ДЛЯ СВОДНОЙ ТАБЛИЦЫ

Данные, необходимые для создания сводной таблицы, должны быть представлены в виде прямоугольной базы данных (прямоугольный диапазон ячеек сводной таблицы). Причем база данных может храниться как в виде диапазона листов (таблица либо обычный диапазон), так и в файле внешней базы данных. Несмотря на то что Excel может генерировать сводные таблицы на основе практически любой базы данных, должны быть соблюдены определенные условия.

Поля в таблице базы данных, на основе которой создается сводная таблица, относятся к следующим двум типам.

- **Данные (Data)**. Включает суммируемые значения. Например, в случае с базой данных о продажах поле `Продажи` является полем данных.
- **Категория (Category)**. Описывает данные. Например, в случае с той же базой данных о продажах поля `Торговый представитель`, `Регион` и `Месяц` — это примеры полей категорий, описывающих данные в поле `Продажи`.

Таблица базы данных, на основе которой может быть построена сводная таблица, должна быть нормализована. Иными словами, каждая запись (или строка) должна включать информацию, описывающую данные.

Единственная таблица базы данных может включать произвольное количество полей данных и категорий. При создании сводной таблицы обычно выполняется суммирование по одному или нескольким полям данных. Ну а значения в полях категорий отображаются в сводной таблице в виде строк, столбцов либо фильтров.

Если изложенная выше концепция не вполне понятна, обратитесь к находящемуся на сайте книги файлу `normalized data.xlsx`. Эта рабочая книга содержит пример диапазона данных до и после выполнения нормализации, являющейся необходимым условием построения сводной таблицы.

Усовершенствование записанного кода сводной таблицы

Как и в случае с большинством записанных макросов, предыдущий пример не настолько эффективен, как следовало бы ожидать. Как уже отмечалось, его выполнение может завершиться ошибкой. В принципе, его желательно упростить, чтобы сделать немного понятнее, а также исключить возможность появления ошибок. Ниже приведен код, переписанный вручную, который создает ту же сводную таблицу, что и макрос из предыдущего раздела.

```
Sub CreatePivotTable()
    Dim PTCache As PivotCache
    Dim PT As PivotTable
```

```

' Создание области кэша
Set PTCache = ActiveWorkbook.PivotCaches.Create( _
    SourceType:=xlDatabase, _
    SourceData:=Range("A1").CurrentRegion)

' Добавление нового листа в сводную таблицу
Worksheets.Add

' Создание сводной таблицы
Set PT = ActiveSheet.PivotTables.Add( _
    PivotCache:=PTCache, _
    TableDestination:=Range("A3"))

' Определение полей
With PT
    .PivotFields("Регион").Orientation = xlPageField
    .PivotFields("Месяц").Orientation = xlColumnField
    .PivotFields("Торговый представитель") _
        .Orientation = xlRowField
    .PivotFields("Продажи").Orientation = xlDataField

    'заголовки полей отсутствуют
    .DisplayFieldCaptions = False
End With
End Sub

```

В данном случае процедура `CreatePivotTable` была упрощена (что облегчило ее понимание) благодаря объявлению двух переменных объекта: `PTCache` и `PT`. Новый объект `PivotCache` был создан с помощью метода `Create`. Также был добавлен рабочий лист, который стал активным (на этом листе размещается сводная таблица). Затем был создан объект `PivotTable` с помощью метода `Add` из коллекции `PivotTables`. В последнем фрагменте кода добавляются поля в сводную таблицу, а также задается их положение в таблице путем присвоения значения свойству `Orientation`.

Обратите внимание на то, что исходный макрос жестко привязан к диапазону данных, на основе которого создается объект `PivotCache` ('Лист1!R1C1:R13C4'), и к местоположению сводной таблицы (Лист2). В процедуре `CreatePivotTable` сводная таблица основана на текущем диапазоне, окружающем ячейку A1. Это гарантирует, что макрос будет выполняться даже тогда, когда в диапазон добавлены дополнительные данные.

Добавление рабочего листа до того, как была создана сводная таблица, исключает необходимость жесткого кодирования ссылки на лист. Еще одно отличие заключается в том, что написанный вручную макрос не определяет имя сводной таблицы. Да это и не требуется, поскольку используется переменная объекта `PT`, которая выполняет эту задачу.



Примечание

Данный код можно сделать более универсальным, используя индексы вместо названий в коллекции `PivotFields`. При этом, если пользователь изменяет заголовки столбцов, код остается работоспособным. Например, в более универсальном макросе используется оператор `PivotFields(1)` вместо оператора `PivotFields('Регион')`.



СОВМЕСТИМОСТЬ СВОДНЫХ ТАБЛИЦ

Если планируется организовать общий доступ к рабочим книгам, включающим сводные таблицы, со стороны пользователей предыдущих версий Excel, уделите особое внимание вопросам совместимости. При просмотре кода записанного макроса (см. раздел “Создание сводной таблицы”) нетрудно заметить следующую инструкцию:

```
DefaultVersion:=xlPivotTableVersion15
```

Если рабочая книга находится в режиме совместимости, на ее месте будет такая инструкция:


```
DefaultVersion:=xlPivotTableVersion10
```

Вы также увидите, что код записанного макроса совершенно отличается. Причина заключается в существенных изменениях в сводных таблицах, начиная с версии Excel 2007.

Предположим, что в среде Excel 2013 создана рабочая книга, которая была передана пользователям Excel 2003. Они увидят таблицу, но не смогут ее обновить. Другими словами, они получат в свое распоряжение статическую таблицу, включающую ряд чисел.

Для обеспечения обратной совместимости сводных таблиц в Excel 2013 следует сначала сохранять файлы в формате XLS, а затем повторно их открывать. В этом случае обеспечивается работоспособность сводных таблиц при открытии содержащих их документов в версиях, предшествующих Excel 2007. Но в этом случае вы не сможете воспользоваться новыми средствами сводных таблиц, которые появились в современных версиях Excel.

К счастью, существует модуль проверки совместимости Excel, который поможет выявить проблемы, связанные с совместимостью (см. следующий рисунок). Но этот модуль не в состоянии проверить поддержку обратной совместимости кода, генерируемого макросом.



Рассматриваемый в этой главе макрос не в состоянии генерировать сводные таблицы, поддерживающие обратную совместимость.

Чтобы лучше понять рассматриваемую тему, запишите собственный макрос и изучите ключевые объекты, методы и свойства. После этого обратитесь к разделам справочного руководства, чтобы разобраться, как все это работает. Практически во всех случаях необходимо вносить изменения в записанный макрос. Как только вам станут понятны принципы управления сводными таблицами, можете приступить к созданию кода без предвзятельной записи макроса.

Создание сложных сводных таблиц

В этом разделе будет представлен код VBA, используемый для создания сравнительно сложной сводной таблицы.

На рис. 15.3 показан фрагмент базы данных на листе. В этой таблице содержится 15840 строк иерархически упорядоченной информации о бюджете компании. В компании существует пять подразделений; каждое подразделение включает одиннадцать отделов. Отдел имеет четыре бюджетные категории, а каждая категория состоит из нескольких пунктов. Бюджетные и фактические расходы указываются для каждого (из двенадцати) месяцев. Цель — создать сводную таблицу, которая суммирует эту информацию.

	A	B	C	D	E	F	G
1	Подразделение	Отдел	Категория	Статья расходов	Месяц	План	Факт
2	Северная Америка	Обработка данных	Компенсации	Зарплата	Янв	2583	3165
3	Северная Америка	Обработка данных	Компенсации	Надбавки	Янв	4496	2980
4	Северная Америка	Обработка данных	Компенсации	Премии	Янв	3768	3029
5	Северная Америка	Обработка данных	Компенсации	Комиссия	Янв	3133	2815
6	Северная Америка	Обработка данных	Компенсации	Налог на фонд оплаты	Янв	3559	3770
7	Северная Америка	Обработка данных	Компенсации	Обучение	Янв	3099	3559
8	Северная Америка	Обработка данных	Компенсации	Конференции	Янв	2931	3199
9	Северная Америка	Обработка данных	Компенсации	Культурно-массовые мероприятия	Янв	2632	2633
10	Северная Америка	Обработка данных	Льготы	Аренда	Янв	2833	2508
11	Северная Америка	Обработка данных	Льготы	Получение платы за аренду	Янв	3450	2631
12	Северная Америка	Обработка данных	Льготы	Коммунальные платежи	Янв	4111	3098
13	Северная Америка	Обработка данных	Льготы	Техническое обслуживание	Янв	3070	2870
14	Северная Америка	Обработка данных	Льготы	Оплата за телефон	Янв	3827	4329
15	Северная Америка	Обработка данных	Льготы	Другие расходы	Янв	3843	3322
16	Северная Америка	Обработка данных	Поставки и обслуживание	Офисные расходы	Янв	2642	3218
17	Северная Америка	Обработка данных	Поставки и обслуживание	Поставки компьютеров	Янв	3052	4098
18	Северная Америка	Обработка данных	Поставки и обслуживание	Книги и программы	Янв	4346	3361
19	Северная Америка	Обработка данных	Поставки и обслуживание	Услуги сторонних организаций	Янв	2869	3717

Рис. 15.3. Данные из этой рабочей книги будут собраны в сводной таблице



На сайте

Рассматриваемая в данном разделе рабочая книга находится на сайте книги (файл budget pivot table.xlsx).

На рис. 15.4 показана сводная таблица, созданная на основе приведенных выше данных. Обратите внимание на то, что она содержит вычисляемое поле, которое называется Отклонение. Значение этого поля представляет собой разницу между значениями полей План и Факт.



Примечание

Можно также вставить новую колонку в таблицу и создать формулу, вычисляющую разницу между плановыми и фактическими показателями. Эта возможность будет недоступной в том случае, если данные берутся из внешнего источника.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	Подразделение	(Все)												
2	Категория	(Все)												
3														
4		Янв	Фев	Мар	Апр	Май	Июн	Июл	Авг	Сен	Окт	Ноя	Дек	Общий итог
5	Бухгалтерия													
6	План	422 455	433 317	420 522	417 964	411 820	414 012	427 431	418 530	412 134	421 678	426 602	418 445	5044 910
7	Факт	422 662	413 163	416 522	420 672	431 303	429 993	425 879	415 253	417 401	417 806	425 271	420 026	5055 951
8	Отклонение	-0 207	20 154	4 000	-2 708	-19 483	-15 981	1 552	3 277	-5 267	3 872	1 331	-1 581	-11 041
9	Коммерческий отдел													
10	План	420 659	421 962	417 814	420 302	422 409	426 802	428 460	420 879	422 334	426 271	408 725	411 322	5047 939
11	Факт	431 565	421 251	408 661	408 912	425 620	428 596	424 737	416 638	408 030	417 463	415 480	413 360	5020 313
12	Отклонение	-10 906	0 711	9 153	11 390	-3 211	-1 794	3 723	4 241	14 304	8 808	-6 755	-2 038	27 626
13	Маркетинг													
14	План	424 590	419 331	417 949	420 324	427 150	424 169	421 183	420 245	429 454	412 078	411 896	423 101	5051 470
15	Факт	416 008	420 828	425 437	417 310	419 996	428 330	428 958	420 856	416 067	419 232	411 739	424 492	5049 253
16	Отклонение	8 582	-1 497	-7 488	3 014	7 154	-4 161	-7 775	-0 611	13 387	-7 154	0 157	-1 391	2 217
17	НИОКР													
18	План	417 771	429 880	424 066	421 539	417 440	421 174	417 151	413 086	417 919	417 782	419 949	419 881	5037 638

Рис. 15.4. Сводная таблица, созданная на основе данных из рис. 15.3

Код сводной таблицы

Ниже приведен код, генерирующий сводную таблицу.

```
Sub CreatePivotTable()
    Dim PTCache As PivotCache
    Dim PT As PivotTable

    Application.ScreenUpdating = False
    ' Удаление листа сводной таблицы (при его наличии)
    On Error Resume Next
    Application.DisplayAlerts = False
    Sheets("Сводная таблица").Delete
    On Error GoTo 0

    ' Создание кеша сводной таблицы
    Set PTCache = ActiveWorkbook.PivotCaches.Create( _
        SourceType:=xlDatabase, _
        SourceData:=Range("A1").CurrentRegion.Address)

    ' Добавление нового рабочего листа
    Worksheets.Add
    ActiveSheet.Name = "Сводная таблица"
    ActiveWindow.DisplayGridlines = False

    ' Создание сводной таблицы на основе данных кеша
    Set PT = ActiveSheet.PivotTables.Add( _
        PivotCache:=PTCache, _
        TableDestination:=Range("A1"), _
        TableName:="BudgetPivot")

    With PT
        ' Добавление полей
        .PivotFields("Категория").Orientation = xlPageField
        .PivotFields("Подразделение").Orientation = xlPageField
        .PivotFields("Отдел").Orientation = xlRowField
        .PivotFields("Месяц").Orientation = xlColumnField
        .PivotFields("План").Orientation = xlDataField
        .PivotFields("Факт").Orientation = xlDataField
        .DataPivotField.Orientation = xlRowField
    End With
End Sub
```



```
'
Добавление вычисляемого поля, определяющего отклонение
.CalculatedFields.Add "Отклонение", "=План-Факт"
.PivotFields("Отклонение").Orientation = xlDataField

'
Определение числового формата
.DataBodyRange.NumberFormat = "0,000"

'
Применение стиля
.TableStyle2 = "PivotStyleMedium2"

'
Скрытие заголовков полей
.DisplayFieldCaptions = False

'
Изменение заголовков
.PivotFields("Сумма по полю План").Caption = " План"
.PivotFields("Сумма по полю Факт").Caption = " Факт"
.PivotFields("Сумма по полю Отклонение").Caption = _
    " Отклонение"
End With
End Sub
```

Принцип работы сводной таблицы

Процедура `CreatePivotTable` начинает свою работу с удаления листа Сводная таблица, если он существует. После этого создается объект `PivotCache`, добавляется новый лист Сводная таблица и создается сводная таблица на основе объекта `PivotCache`. Далее программа добавляет следующие поля к созданной сводной таблице:

- **Категория** — поле фильтра отчета (страницы).
- **Подразделение** — поле фильтра отчета (страницы).
- **Отдел** — поле строки.
- **Месяц** — поле столбца.
- **План** — поле данных.
- **Факт** — поле данных.

Обратите внимание на то, что свойство `Orientation` объекта `DataPivotField` установлено равным `xlRowField` с помощью следующего оператора:

```
.DataPivotField.Orientation = xlRowField
```

Этот оператор определяет общую ориентацию сводной таблицы и представляет поле в области ЗНАЧЕНИЯ в списке полей сводной таблицы (рис. 15.5). Попробуйте переместить это поле в область КОЛОННЫ, после чего посмотрите, как изменится макет сводной таблицы.

Затем процедура использует метод `Add` из коллекции `CalculatedFields` для создания вычисляемого поля `Отклонение`, величина которого представляет собой результат вычитания значения поля `Факт` из значения поля `План`. Это вычисляемое поле относится к категории полей данных.



Примечание

Чтобы добавить вычисляемое поле в сводную таблицу вручную, воспользуйтесь командой Работа со сводными таблицами⇒Анализ⇒Вычисления⇒Поля, элементы и наборы⇒Вычисляемое поле (PivotTable Tools⇒Analysis⇒Calculations⇒Fields, Items & Sets⇒Calculation Field) для перехода в диалоговое окно Вставка вычисляемого поля (Insert Calculated Field).




Рис. 15.5. Список полей сводной таблицы

Также код выполняет небольшие “косметические” улучшения сводной таблицы.

- Применяет числовой формат к объекту `DataBodyRange` (этот объект представляет данные всей сводной таблицы).
- Применяет стиль.
- Скрывает заголовки, что эквивалентно команде **Работа со сводными таблицами** ⇒ **Анализ** ⇒ **Показать/Скрыть** ⇒ **Заголовки полей** (`PivotTable Toos` ⇒ `Analysis` ⇒ `Show/Hide` ⇒ `Field Headers`).
- Изменяет заголовки, отображенные в сводной таблице. Например, название *Сумма по полю План* (`Sum of Budget`) заменяется названием *План* (`Budget`). Обратите внимание на то, что перед названием *План* вставляется пробел. В Excel не допускается изменение заголовка, который соответствует имени поля, но, как видите, это ограничение можно обойти, добавив пробел.



Примечание

В процессе создания процедуры я постоянно пользовался средством записи макросов, чтобы изучить различные свойства. Также мне очень помогла интерактивная справочная система (ну и многие вещи пришлось делать методом проб и ошибок).

Создание нескольких сводных таблиц

При выполнении данного упражнения создается целая серия сводных таблиц, в которых суммируются данные, собранные в процессе опроса покупателей. Эти данные находятся в базе данных рабочих листов (рис. 15.6), которая состоит из 150 строк. Каждая

строка включает указание пола респондента, а также числовой рейтинг, который изменяется от 1 до 5 по каждому из 14 пунктов опроса.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P
1	Имя	Пол	Удобство расположения склада	Удобное время работы	Качество поддержки	Быстрота дозвона	Практичность сайта	Вежливость сотрудников	Готовность прийти на помощь	Эрудированность сотрудников	Конкурентность цен	Богатство ассортимента	Привлекательность рекламы на ТВ	Качество продукции	В целом я доволен	Рекомендую вашу компанию
2	Вопрос1	Мужской	1	4	4	4	1	1	2	1	1	2	5	2	2	1
3	Вопрос2	Женский	2	5	1	1	4	2	4	3	3	2	2	5	2	3
4	Вопрос3	Мужской	1	1	4	2	3	3	2	1	2	3	2	4	3	2
5	Вопрос4	Мужской	2	1	3	5	1	2	3	4	2	1	3	4	1	2
6	Вопрос5	Женский	2	2	5	5	4	2	1	5	5	2	3	4	2	5
7	Вопрос6	Женский	2	4	3	3	1	1	4	4	4	2	2	2	2	4
8	Вопрос7	Женский	2	4	5	4	5	3	2	5	4	4	1	5	4	4
9	Вопрос8	Мужской	3	2	1	2	3	4	3	1	2	4	3	4	4	2
10	Вопрос9	Женский	3	4	4	4	5	1	4	1	4	1	2	1	1	4
11	Вопрос10	Мужской	2	1	5	5	5	1	4	1	2	2	5	2	2	2
12	Вопрос11	Мужской	4	3	3	2	1	2	4	2	1	4	2	2	4	1
13	Вопрос12	Женский	2	1	4	5	5	3	1	4	1	2	3	4	4	4
14	Вопрос13	Женский	4	3	4	3	2	5	3	3	2	2	5	2	4	2
15	Вопрос14	Женский	2	3	4	2	1	1	4	2	1	3	3	1	3	1
16	Вопрос15	Женский	1	3	5	1	2	2	4	1	3	4	2	5	4	3
17	Вопрос16	Мужской	1	4	1	3	4	3	4	4	5	3	4	1	3	3
18	Вопрос17	Женский	3	4	3	5	5	4	4	3	2	4	2	2	4	2
19	Вопрос18	Мужской	1	5	5	3	5	3	4	2	3	2	3	3	2	3
20	Вопрос19	Женский	1	3	5	4	5	5	5	1	1	5	3	2	5	1
21	Вопрос20	Мужской	2	2	5	2	2	5	5	3	1	5	2	4	5	1
22	Вопрос21	Мужской	3	4	1	4	5	1	3	1	4	1	2	1	1	4
23	Вопрос22	Мужской	2	1	5	5	5	1	2	1	2	2	5	2	2	2
24	Вопрос23	Мужской	4	3	4	2	1	2	1	2	1	4	4	1	4	2
25	Вопрос24	Женский	1	1	2	5	5	5	3	1	4	1	2	3	1	2

Рис. 15.6. Создание набора сводных таблиц, суммирующих данные опроса



На сайте

Эта рабочая книга находится на сайте книги (файл `survey data pivot tables.xlsx`).

На рис. 15.7 показаны некоторые из 28 сводных таблиц, созданные макросом. Суммирование по каждому элементу опроса производилось с помощью двух сводных таблиц (в одной отображаются фактические результаты опроса, в другой — процент респондентов, ответивших на пункты опроса).

	A	B	C	D	E	F	G	H	I
1	Удобство расположения склада				Удобство расположения склада				
2	Количество по полю Удобство расположения склада				Количество по полю Удобство расположения склада				
3		Женский	Мужской	Общий итог		Женский	Мужской	Общий итог	
4	Решительно не согласен	28	40	68	Решительно не согласен	39%	51%	45%	
5	Не согласен	20	16	36	Не согласен	28%	20%	24%	
6	Коллеблюсь	15	9	24	Коллеблюсь	21%	11%	16%	
7	Согласен	6	14	20	Согласен	08%	18%	13%	
8	Полностью согласен	2	2	2	Полностью согласен	03%	00%	01%	
9	Общий итог	71	79	150					
10	Удобное время работы				Удобное время работы				
11	Количество по полю Удобное время работы				Количество по полю Удобное время работы				
12		Женский	Мужской	Общий итог		Женский	Мужской	Общий итог	
14	Решительно не согласен	11	13	24	Решительно не согласен	15%	16%	16%	
15	Не согласен	7	11	18	Не согласен	10%	14%	12%	
16	Коллеблюсь	30	26	56	Коллеблюсь	42%	33%	37%	
17	Согласен	20	22	42	Согласен	28%	28%	28%	
18	Полностью согласен	3	7	10	Полностью согласен	04%	09%	07%	
19	Общий итог	71	79	150					
20	Качество поддержки				Качество поддержки				
21	Количество по полю Качество поддержки				Количество по полю Качество поддержки				
22		Женский	Мужской	Общий итог		Женский	Мужской	Общий итог	
24	Решительно не согласен	7	14	21	Решительно не согласен	10%	18%	14%	
25	Не согласен	7	4	11	Не согласен	10%	05%	07%	
26	Коллеблюсь	16	14	30	Коллеблюсь	23%	18%	20%	
27	Согласен	20	20	40	Согласен	28%	25%	26%	
28	Общий итог	71	79	150					

Рис. 15.7. Некоторые из 28 сводных таблиц, созданных с помощью процедуры VBA

Ниже приводится процедура VBA, с помощью которой создавались сводные таблицы.

```
Sub MakePivotTables()  
    ' Эта процедура создает 28 сводных таблиц  
    Dim PTCache As PivotCache  
    Dim PT As PivotTable  
    Dim SummarySheet As Worksheet  
    Dim ItemName As String  
    Dim Row As Long, Col As Long, i As Long  
  
    Application.ScreenUpdating = False  
  
    ' Удаление листа итогов при его наличии  
    On Error Resume Next  
    Application.DisplayAlerts = False  
    Sheets("Итог").Delete  
    On Error GoTo 0  
  
    ' Добавление листа итогов  
    Set SummarySheet = Worksheets.Add  
    ActiveSheet.Name = "Итог"  
  
    ' Создание кеша сводной таблицы  
    Set PTCache = ActiveWorkbook.PivotCaches.Create( _  
        SourceType:=xlDatabase, _  
        SourceData:=Sheets("ДанныеОпроса").Range("A1"). _  
        CurrentRegion)  
  
    Row = 1  
    For i = 1 To 14  
        For Col = 1 To 6 Step 5 '2 колонки  
            ItemName = Sheets("ДанныеОпроса").Cells(1, i + 2)  
            With Cells(Row, Col)  
                .Value = ItemName  
                .Font.Size = 16  
            End With  
  
            ' Создание сводной таблицы  
            Set PT = ActiveSheet.PivotTables.Add( _  
                PivotCache:=PTCache, _  
                TableDestination:=SummarySheet.Cells( _  
                    Row + 1, Col))  
  
            ' Добавление полей  
            If Col = 1 Then 'таблицы частот  
                With PT.PivotFields(ItemName)  
                    .Orientation = xlDataField  
                    .Name = "Частота"  
                    .Function = xlCount  
                End With  
            Else ' процентные таблицы  
                With PT.PivotFields(ItemName)  
                    .Orientation = xlDataField  
                    .Name = "Процент"  
                    .Function = xlCount  
                End With  
            End If  
        Next Col  
    Next i  
End Sub
```

```
        .Calculation = xlPercentOfColumn
        .NumberFormat = "0,0%"
    End With
End If
PT.PivotFields(ItemName).Orientation = xlRowField
PT.PivotFields("Пол").Orientation = xlColumnField
PT.TableStyle2 = "PivotStyleMedium2"
PT.DisplayFieldCaptions = False
If Col = 6 Then
    ' добавление шкал данных в последнюю колонку
    PT.ColumnGrand = False
    PT.DataBodyRange.Columns(3).FormatConditions. _
        AddDatabar
    With pt.DataBodyRange.Columns(3). _
        FormatConditions(1)
        .BarFillType = xlDataBarFillSolid
        .MinPoint.Modify newtype:= _
            xlConditionValueNumber, newvalue:=0
        .MaxPoint.Modify newtype:= _
            xlConditionValueNumber, newvalue:=1
    End With
End If
Next Col
Row = Row + 10
Next i

' Замена чисел описательным текстом
With Range("A:A,F:F")
    .Replace "1", "Решительно не согласен"
    .Replace "2", "Не согласен"
    .Replace "3", "Колеблюсь"
    .Replace "4", "Согласен"
    .Replace "5", "Полностью согласен"
End With
End Sub
```

Обратите внимание на то, что все сводные таблицы были созданы на основе единственного объекта `PivotCache`.

Сводные таблицы создаются с помощью вложенных циклов. Значение счетчика цикла `Col` изменяется от 1 до 6 с помощью параметра `Step`. Для второго столбца сводных таблиц немного изменяется характер обработки. Выполняются следующие действия:

- отображается величина (в виде процента от значения в столбце);
- не отображаются окончательные итоги для строк;
- используется числовой формат;
- отображаются цветовые шкалы, с помощью которых реализуется условное форматирование.

Переменная `Row` отслеживает начальную строку в каждой сводной таблице. Завершающая операция заключается в замене числовых категорий в столбцах А и F текстом. Например, вместо единицы подставляется текст “Решительно не согласен”.

Создание обратной сводной таблицы

Сводная таблица представляет собой результат суммирования данных в обычной таблице. А как быть в случае, если у вас имеется итоговая таблица, на основе которой нужно воссоздать исходную таблицу? Соответствующий пример приводится на рис. 15.8. Диапазон ячеек B2:F14 представляет собой итоговую таблицу — упрощенный вариант сводной таблицы. В столбцах I:K находится состоящая из 48 строк таблица, созданная на основе итоговой таблицы. В этой таблице каждая строка содержит точку данных, а первые два столбца включают описание этой точки.

	A	B	C	D	E	F	G	H	I	J	K
1									Столбец1	Столбец2	Столбец3
2		Месяц	Эми	Боб	Чак	Дуг			Янв	Эми	47 955
3		Янв	47 955	34 240	55 560	56 380			Янв	Боб	34 240
4		Фев	44 715	35 435	61 810	63 325			Янв	Чак	55 560
5		Мар	41 635	34 005	58 655	60 055			Янв	Дуг	56 380
6		Апр	48 515	32 065	63 530	57 700			Фев	Эми	44 715
7		Май	53 945	39 225	67 860	57 900			Фев	Боб	35 435
8		Июн	50 990	38 305	64 370	61 760			Фев	Чак	61 810
9		Июл	49 235	38 675	66 020	65 220			Фев	Дуг	63 325
10		Авг	55 725	34 300	70 160	63 140			Мар	Эми	41 635
11		Сен	57 710	26 615	68 985	65 740			Мар	и	34 005
12		Окт	54 020	24 220	70 035	63 300			Мар	Чак	58 655
13		Ноя	52 055	19 365	65 240	62 905			Мар	Дуг	60 055
14		Дек	48 690	20 440	64 165	54 915			Апр	Эми	48 515
15									Апр	Боб	32 065
16									Апр	Чак	63 530
17									Апр	Дуг	57 700
18									Май	Эми	53 945
19									Май	Боб	39 225
20									Май	Чак	67 860
21									Май	Дуг	57 900
22									Июн	Эми	50 990
23									Июн	Боб	38 305
24									Июн	Чак	64 370
25									Июн	Дуг	61 760

Рис. 15.8. Итоговая таблица (слева) может быть преобразована в обычную (справа)

В Excel не существует прямого способа преобразования итоговой таблицы в обычную, но эту работу может выполнить макрос VBA. После его создания я разработал диалоговое окно UserForm, показанное на рис. 15.9. В нем определяются входной и выходной диапазоны, а также имеется опция преобразования выходного диапазона в таблицу.



На сайте

Эта рабочая книга находится на сайте книги (файл `reverse pivot table.xlsx`).

После щелчка мышью на кнопке ОК в диалоговом окне UserForm код VBA проверяет диапазоны и вызывает процедуру `ReversePivot` с помощью следующего оператора:

```
Call ReversePivot(SummaryTable, OutputRange, cbCreateTable)
```

Месяц	Эми	Боб	Чак	Дуг
Янв	47 955	34 240	55 560	56 380
Фев	44 715	35 435	61 810	63 325
Мар	41 635	34 005	58 655	60 055
Апр	48 515	32 065	63 530	57 700
Май	53 945	39 225	67 860	57 900
Июн	50 990	38 305	64 370	61 760
Июл	49 235	38 675	66 020	65 220
Авг	55 725	34 300	70 160	63 140
Сен	57 710	26 615	68 985	65 740
Окт	54 020	24 220	70 035	63 300
Ноя	52 055	19 365	65 240	62 905
Дек	48 690	20 440	64 165	54 915

Обращение сводной таблицы

Создание 3-колоночной таблицы на основе таблицы итогов.

Месяц	Эми	Боб	Чак	Дуг
Янв	13,040	10,014	9,880	10,000
Фев	13,130	10,412	18,940	19,440
Мар	12,640	10,804	11,810	12,650
Апр	12,040	10,898	10,990	11,810
Май	14,970	10,920	10,960	10,040
Июн	14,110	10,720	10,960	10,880
Июл	14,110	10,720	10,960	10,880
Авг	14,970	10,920	10,960	10,040
Сен	14,970	10,920	10,960	10,040
Окт	14,970	10,920	10,960	10,040
Ноя	14,970	10,920	10,960	10,040
Дек	14,970	10,920	10,960	10,040

Диапазон ячеек: \$B\$2:\$F\$14

Выходной диапазон (одна ячейка): Данные!\$H\$2

Форматированная таблица

Отмена OK

Месяц	Эми	Боб	Чак	Дуг
Май	Чак			67 860
Мвй	Дуг			57 900
Июн	Эми			50 990
Июн	Боб			38 305
Июн	Чак			64 370
Июн	Дуг			61 760

Рис. 15.9. В этом диалоговом окне пользователь может определить диапазоны

Этой процедуре передаются следующие три аргумента.

- SummaryTable. Объект Range, который представляет итоговую таблицу.
- OutputRange. Объект Range, представляющий левую верхнюю ячейку выходного диапазона.
- cbCreateTable. Флажок (объект Checkbox) в окне UserForm.

Эта процедура работает с итоговой таблицей любого размера. Количество строк данных в выходной таблице можно подсчитать по формуле $(r-1) * (c-1)$, в которой переменные r и c представляют собой количество строк и столбцов в итоговой таблице соответственно.

Ниже приведен код процедуры ReversePivot.

```
Sub ReversePivot(SummaryTable As Range, _
    OutputRange As Range, CreateTable As Boolean)
    Dim r As Long, c As Long
    Dim OutRow As Long, OutCol As Long
```

```
    ' Преобразование диапазона
    OutRow = 2
    Application.ScreenUpdating = False
    OutputRange.Range("A1:C3") = Array("Столбец", _
        "Столбец2", "Столбец3")
    For r = 2 To SummaryTable.Rows.Count
```

```
For c = 2 To SummaryTable.Columns.Count
    OutputRange.Cells(OutRow, 1) = SummaryTable.Cells(r, 1)
    OutputRange.Cells(OutRow, 2) = SummaryTable.Cells(1, c)
    OutputRange.Cells(OutRow, 3) = SummaryTable.Cells(r, c)
    OutRow = OutRow + 1
Next c
Next r

' Создать таблицу?
If CreateTable Then _
    ActiveSheet.ListObjects.Add xlSrcRange, _
        OutputRange.CurrentRegion, , xlYes
End Sub
```

Описанная процедура достаточно проста. Код выполняет циклический обход строк и столбцов во входном диапазоне, а затем записывает данные в выходной диапазон, который всегда состоит из трех столбцов. Переменная `OutRow` отслеживает текущую строку в выходном диапазоне. Если пользователь устанавливает флажок, выходной диапазон преобразуется в таблицу. При этом используется метод `Add` из коллекции `ListObjects`.